

Integrity Check for the Epidemiology of Diabetes Interventions and Complications Study (EDIC) Adolescent Four Year Retinopathy, Nephropathy and Hypoglycemia Datasets

As a partial check of the integrity of the EDIC Adolescent Four Year Retinopathy, Nephropathy and Hypoglycemia datasets archived in the NIDDK data repository, a set of tabulations was performed to verify that published results can be reproduced using the archived datasets. Analyses were performed to duplicate published results for the data reported by the DCCT/EDIC Research Group [1] in *The Journal of Pediatrics* in December 2001. The results of this integrity check are described below. The full text of *The Journal of Pediatrics* article can be found in Attachment 1, and the SAS code for our tabulations is included in Attachment 2.

The intent of this DSIC is to provide confidence that the data distributed by the NIDDK repository is a true copy of the study data. Our intent is *not* to assess the integrity of the statistical analyses reported by study investigators. As with all statistical analyses of complex datasets, complete replication of a set of statistical results should not be expected on a first (or second) exercise in secondary analysis. This occurs for a number of reasons including differences in the handling of missing data, restrictions on cases included in samples for a particular analysis, software coding used to define complex variables, etc. Experience suggests that most discrepancies can ordinarily be resolved by consultation with the study data coordinating center (DCC), however this process is labor-intensive for both DCC and Repository staff. It is thus not our policy to resolve every discrepancy that is observed in an integrity check. Specifically, we do not attempt to resolve minor or inconsequential discrepancies with published results or discrepancies that involve complex analyses, *unless staff of the NIDDK Repository suspect that the observed discrepancy suggests that the dataset may have been corrupted in storage, transmission, or processing by repository staff*. We do, however, document in footnotes to the integrity check those instances in which our secondary analyses produced results that were not fully consistent with those reported in the target publication.

Background. The Diabetes Control and Complications Trial (DCCT), 1983-1993, was a multi-center trial designed to determine whether intensive therapy to maintain blood glucose and glycosolated hemoglobin concentrations as close to the normal range as possible would prevent or delay long-term complications in patients with Type 1 diabetes. This trial showed a markedly reduced risk of microvascular complications as compared with conventional therapy. Most participants were then enrolled in the EDIC study, a long-term observational study. One of the objectives of the EDIC study was to compare the long-term effects of the intensive or conventional therapy provided during the DCCT on the development of more advanced retinal and renal complications of diabetes. The published data describes differences between the two original treatment groups in the incidence of these complications four years after the close of the DCCT, for the adolescent DCCT cohort [1].

In summary, in 1994, 91% (n=175) of the surviving DCCT adolescents (n=193) were enrolled in EDIC for regular observational follow-up of metabolic and complications status. Diabetes care was obtained from the EDIC participants' own physicians [2].

Baseline Data. Table 1 of the 2001 *Journal of Pediatrics* article reports on baseline characteristics. Variables summarized in this baseline table (Table 1. Characteristics of EDIC patients who were adolescents at entry into the DCCT and who had an HbA1c at year 4 of EDIC) are taken from the EDICADO4 analysis dataset created for this study and the DCCT baseline dataset, BASELINE. Table A lists the variables used for replication of the Table 1 variables.

Table A: Variables Used to Replicate Table 1

Table 1 Variable	Variables Used in Replication
Sample size	EDICADO4 dataset: group (if hba1c was measured at EDIC year 4)
Age at EDIC year 4 (y)	EDICADO4 dataset: att_age
Primary prevention cohort (%)	BASELINE dataset: rebase (where rebase='PRIM')
Duration of type 1 diabetes at EDIC year 4 (y)	EDICADO4 dataset: att_dur
Duration since entry into DCCT (y)	EDICADO4 dataset: in_study
Female (%)	EDICADO4 dataset: sex (where sex='F')
Women pregnant during EDIC (%)	EDICADO4 dataset: tot_preg (if tot_preg>0)
Women pregnant at EDIC year 4, n (%)	EDICADO4 dataset: pregnant
Insulin regimen at EDIC year 4	EDICADO4 dataset: obinsreg
1-2 injections (%)	(where obinsreg=3)
MDI ⁽¹⁾	(where obinsreg=2)
CSII ⁽¹⁾	(where obinsreg=1)
Total daily dose (U/kg/day) at EDIC year 4	EDICADO4 dataset: std_ins
SBMG ⁽¹⁾ \geq 4/day (%) at EDIC year 4	EDICADO4 dataset: sbgm_4
HbA1c (%)	EDICADO4 dataset: hba1c
Notes: (1) MDI, multiple daily injections; CSII, continuous subcutaneous insulin infusion; SMBG, self-monitoring blood glucose	

In Table B, we compare the results for characteristics calculated from the archived dataset to the results published in the 2001 *Journal of Pediatrics* article. As Table B shows, the results obtained from the archived data are the same as those in the published tabulations, with a few minor exceptions. Also note, the p-value for 'Insulin regimen at EDIC year 4' was calculated as 0.0022, which agrees with the published result of '<.01'.

Table B: Comparison of Baseline Table Values Computed in Integrity Check to Reference Article Values

Table 1 Variable	Treatment: Intensive ⁽¹⁾		
	<i>Journal of Pediatrics (2001)</i>	Integrity Check	Difference
Sample size	81	81	0
Age at EDIC year 4 (y)	27.2 ± 2.4	27.2 ± 2.4	0
Primary prevention cohort (%)	57	57	0
Duration of type 1 diabetes at EDIC year 4 (y)	17.4 ± 4.2	17.4 ± 4.2	0
Duration since entry into DCCT (y)	12.0 ± 1.9	12.0 ± 1.9	0
Female (%)	48	48	0
Women pregnant during EDIC (%)	26	26	0
Women pregnant at EDIC year 4, n (%)	4 (10)	4 (10)	0
Insulin regimen at EDIC year 4 ⁽²⁾			
1-2 injections (%)	10	9	1
MDI ⁽³⁾	69	69	0
CSII ⁽³⁾	22	23	1
Total daily dose (U/kg/day) at EDIC year 4	.82 ± 0.28	.81 ± 0.28	.01 ± 0
SBMG ⁽³⁾ ≥ 4/day (%) at EDIC year 4	24	27	3
HbA1c (%)	8.4 ± 1.7	8.4 ± 1.7	0
Notes:			
(1) Values are mean ± SD unless otherwise indicated			
(2) P-value <0.1 calculated on the basis of χ^2 test, intensive group versus conventional group			
(3) MDI, multiple daily injections; CSII, continuous subcutaneous insulin infusion; SMBG, self-monitoring blood glucose			

Table B: Comparison of Baseline Table Values Computed in Integrity Check to Reference Article Values (cont.)

Table 1 Variable	Treatment: Conventional ⁽¹⁾		
	<i>Journal of Pediatrics</i> (2001)	Integrity Check	Difference
Sample size	89	89	0
Age at EDIC year 4 (y)	26.3 ± 2.2	26.3 ± 2.2	0
Primary prevention cohort (%)	67	67	0
Duration of type 1 diabetes at EDIC year 4 (y)	16.3 ± 4.3	16.3 ± 4.3	0
Duration since entry into DCCT (y)	11.3 ± 1.9	11.3 ± 1.9	0
Female (%)	60	60	0
Women pregnant during EDIC (%)	36	36	0
Women pregnant at EDIC year 4, n (%)	9 (17)	9 (17)	0
Insulin regimen at EDIC year 4 ⁽²⁾			
1-2 injections (%)	31	30	1
MDI ⁽³⁾	52	53	1
CSII ⁽³⁾	17	17	0
Total daily dose (U/kg/day) at EDIC year 4	.80 ± 0.22	.80 ± 0.22	0
SBMG ⁽³⁾ ≥ 4/day (%) at EDIC year 4	29	31	2
HbA1c (%)	8.4 ± 1.6	8.4 ± 1.6	0
Notes:			
(1) Values are mean ± SD unless otherwise indicated			
(2) P-value <0.1 calculated on the basis of χ^2 test, intensive group versus conventional group			
(3) MDI, multiple daily injections; CSII, continuous subcutaneous insulin infusion; SMBG, self-monitoring blood glucose			

Hypoglycemia Data. Table 2 of the 2001 *Journal of Pediatrics* article reports on rates of hypoglycemia. Variables summarized in this follow-up table (Table 2. Summary of rates of hypoglycemia in former DCCT adolescents during the DCCT and the first 4 years of EDIC) are taken from all three analysis datasets created for this study, EDICADO4, EDAD4HYP and EDAD4ADL. Table C lists the variables used for replication of the Table 2 variables.

The DCC has noted that Table 2 of the 2001 *Journal of Pediatrics* article reports event rates that are weighted by follow-up time. The formula used is:

$$\text{Mean event rate} = (\text{total number of events} / \text{total number of follow-up days}) * 365.25 * 100$$

Further details on the derivation of this formula may be referenced in Biostatistical Methods: The assessment of relative risks, Chapter 8 [3].

Additionally, the DCC has noted that two variables on the EDIC group, hypoglycemia dataset (EDAD4ADL dataset: EDCSRATE, EDRARATE) were not reported in the correct unit of 100 patient years. These variables were, instead, reported based on 100 patient days.

Table C: Variables Used to Replicate Table 2

Table 2 Variable	Variables Used in Replication
Sample size	EDAD4HYP dataset: group EDAD4ADL dataset: group (where teen=1) EDICADO4 dataset : hba1c (where hba1c > 0)
Coma or seizure events	EDAD4HYP dataset: danycs EDAD4ADL dataset: edanycs
Coma or seizure event rate per 100 patient-years	EDAD4HYP dataset: dtcs, dcsday EDAD4ADL dataset: edcs, edmon
All events requiring assistance (including coma and seizure)	EDAD4HYP dataset: danyra EDAD4ADL dataset: edanyra
All events requiring assistance rate per 100 patient-years	EDAD4HYP dataset: dtra, draday EDAD4ADL dataset: edra, edmon

In Table D, we compare the follow-up results calculated from the archived dataset to the results published in the 2001 *Journal of Pediatrics* article. As Table D shows, the results obtained from the archived data are the same as those in the published tabulations, with a few minor exceptions.

Table D: Comparison of Hypoglycemia Table Values Computed in Integrity Check to Reference Article Values

Table 2 Variable	DCCT Results ⁽¹⁾		
	<i>Journal of Pediatrics (2001)</i>	Integrity Check	Difference
Sample size			
INT ⁽²⁾	92	92	0
CON ⁽²⁾	103	103	0
Coma or seizure			
Having events, %			
INT	63	63	0
CON	25	25	0
Event rate/100 patient-years			
INT	26.7	28.2	1.5
CON	9.7	9.5	0.2
RR ⁽²⁾ (INT vs CON)	2.93	2.50	0.43
P-value	< .001	< .001	0
All events requiring assistance (including coma and seizure)			
Having events, %			
INT	82	82	0
CON	45	45	0
Event rate/100 patient-years			
INT	85.7	87.2	1.5
CON	27.8	28.1	0.3
RR (INT vs CON)	2.96	1.83	1.13
P-value	< .001	< .001	0
Notes:			
(1) Events reported at quarterly visits during the DCCT and during EDIC at annual visits reflecting the previous 3 months			
(2) CON, Conventional; INT, intensive; RR, relative risk			

Table D: Comparison of Hypoglycemia Table Values Computed in Integrity Check to Reference Article Values (cont.)

Table 2 Variable	EDIC Results ⁽¹⁾		
	<i>Journal of Pediatrics (2001)</i>	Integrity Check	Difference
Sample size			
INT ⁽²⁾	81	81	0
CON ⁽²⁾	89	89	0
Coma or seizure			
Having events, %			
INT	9.0	10.0	1.0
CON	14.7	15.7	1.0
Event rate/100 patient-years			
INT	16.6	19.2	2.6
CON	26.8	31.8	5.0
RR ⁽²⁾ (INT vs CON)	0.62	0.64	0.02
P-value	.358	.270	.088
All events requiring assistance (including coma and seizure)			
Having events, %			
INT	24.7	27.5	2.8
CON	24.2	25.8	1.6
Event rate/100 patient-years			
INT	51.0	60.0	9.0
CON	57.0	62.2	5.2
RR (INT vs CON)	.90	1.06	0.16
P-value	.749	.808	.059
Notes:			
(1) Events reported at quarterly visits during the DCCT and during EDIC at annual visits reflecting the previous 3 months			
(2) CON, Conventional; INT, intensive; RR, relative risk			

Retinopathy Progression Data. Table 3 of the 2001 *Journal of Pediatrics* article reports on further progression of retinopathy. Variables summarized in this follow-up table (Table 3. Further progression of at least 3 steps in retinopathy level in former DCCT adolescents between DCCT closeout and the EDIC 4-year visit) are taken from the EDICADO4 analysis dataset created for this study. Table E lists the variables used for replication of the Table 3 variables.

Table E: Variables Used to Replicate Table 3

Table 3 Variable	Variables Used in Replication
Sample size	group
Retinopathy worsened by ≥ 3 steps during EDIC	new3step
No retinopathy (10/10)	subgrpr (where subgrpr=1)
Microaneurysms only ($20/\leq 20$)	subgrpr (where subgrpr=2)
Mild NPDR ($35/\leq 35$)	subgrpr (where subgrpr=3)
\geq Moderate NPDR ($\geq 43/< 43$)	subgrpr (where subgrpr=4)

In Table F, we compare the follow-up results calculated from the archived dataset to the results published in the 2001 *Journal of Pediatrics* article. As Table F shows, the results obtained from the archived data are the same as those in the published tabulations, with one minor exception.

Table F: Comparison of Progression of Retinopathy Table Values Computed in Integrity Check to Reference Article Values

Table 3 Variable	Number of Patients ⁽¹⁾		
	<i>Journal of Pediatrics (2001)</i>	Integrity Check	Difference
All patients			
CON ⁽²⁾	71	71	0
INT ⁽²⁾	70	70	0
DCCT closeout status			
No retinopathy (10/10)			
CON	6	6	0
INT	15	15	0
Microaneurysms only (20/≤20)			
CON	27	27	0
INT	28	28	0
Mild NPDR ⁽²⁾ (35/≤35)			
CON	21	21	0
INT	22	22	0
≥Moderate NPDR (≥43/<43)			
CON	17	17	0
INT	5	5	0
Notes:			
(1) Patients with scatter photocoagulation during the DCCT are excluded (4 in the conventional group, 2 in the intensive group)			
(2) CON, conventional; INT, intensive; NPDR, Nonproliferative diabetic retinopathy			

Table F: Comparison of Progression of Retinopathy Table Values Computed in Integrity Check to Reference Article Values (cont.)

Table 3 Variable	Progression of Retinopathy ⁽¹⁾ (%)		
	<i>Journal of Pediatrics (2001)</i>	Integrity Check	Difference
All patients			
CON ⁽²⁾	25.4	25.4	0
INT ⁽²⁾	7.1	7.1	0
DCCT closeout status			
No retinopathy (10/10)			
CON	33.3	33.3	0
INT	0.0	0.0	0
Microaneurysms only (20/≤20)			
CON	11.1	11.1	0
INT	3.6	3.6	0
Mild NPDR ⁽²⁾ (35/≤35)			
CON	19.1	19.1	0
INT	9.1	9.1	0
≥Moderate NPDR (≥43/<43)			
CON	52.6	52.9	0.3
INT	40.0	40.0	0
Notes:			
(1) Percent of patients with stated retinopathy status at DCCT closeout who worsened by ≥3 steps during EDIC. Patients with scatter photocoagulation since the DCCT counted as worse for ≥3-steps progression of retinopathy.			
(2) CON, conventional; INT, intensive; NPDR, Nonproliferative diabetic retinopathy			

Table F: Comparison of Progression of Retinopathy Table Values Computed in Integrity Check to Reference Article Values (cont.)

Table 3 Variable	Odds Reduction (95% CL ⁽¹⁾), p-value		
	<i>Journal of Pediatrics</i> (2001)	Integrity Check	Difference
All patients	77 (39,92), p=.004	77 (35,92), p=.004	0 (4,0), 0
DCCT closeout status			
No retinopathy (10/10)	94 (NC ⁽¹⁾)	100 (NC)	6
Microaneurysms only (20/≤20)	70 (NC)	70 (NC)	0
Mild NPDR ⁽¹⁾ (35/≤35)	56 (NC)	58 (NC)	2
≥Moderate NPDR (≥43/<43)	41 (NC)	41 (NC)	0
Stratified-adjusted ⁽²⁾	67 (8,88), p=.034	70 (6,90), p=.039	3 (2,2), .005
Notes:			
(1) CL, Confidence limit; NC, not computed (because of small size of stratum); NPDR, Nonproliferative diabetic retinopathy			
(2) From the Mantel-Haenzel method			

Other Results. Remaining figures and follow-up tables from the 2001 *Journal of Pediatrics* article were spot checked and found to be consistent with replicated results.

The few discrepancies documented in this report are likely due to data corrections and updates made between the paper data freeze and the final data freeze. The repository has high confidence in the integrity of the Retinopathy, Nephropathy and Hypoglycemia analysis datasets.

Notes

1. All three analysis datasets related to the EDIC Adolescent Four Year Retinopathy, Nephropathy and Hypoglycemia follow-up were examined in this replication analysis. In addition to the analysis datasets, the DCCT baseline dataset, which is housed at the repository, should be used to identify the primary prevention cohort.

References

1. DCCT/EDIC Research Group. Writing Committee: Neil H. White, MD, Patricia A. Cleary, MS, William Dahms, MD, David Goldstein, MD, John Malone, MD, and William V. Tamborlane, MD, **Beneficial effects of intensive therapy of diabetes during adolescence: Outcomes after the conclusion of the Diabetes Control and Complications Trial (DCCT)**, The Journal of Pediatrics, December 2001; Volume 139, Number 6: 804-812.
2. NIDDK Website: EDIC page. [NIDDK : Epidemiology of Diabetes Interventions and Complications Study \(EDIC\)](#)
3. Lachin JM., Biostatistical Methods: The assessment of relative risks. John Wiley and Sons; 2000.

ATTACHMENT 1

Full Text of Article

DCCT/EDIC Research Group. Writing Committee: Neil H. White, MD, Patricia A. Cleary, MS, William Dahms, MD, David Goldstein, MD, John Malone, MD, and William V. Tamborlane, MD, Beneficial effects of intensive therapy of diabetes during adolescence: Outcomes after the conclusion of the Diabetes Control and Complications Trial (DCCT), The Journal of Pediatrics, December 2001; Volume 139, Number 6: 804-812.

NOTE. Single copies of articles published in scientific journals are included with this documentation. These articles are copyrighted, and the repository has purchased ONE reprint from their publisher to include with this documentation. If additional copies are made of these copyrighted articles, users are advised that payment is due to the copyright holder (typically the publisher of the scientific journal).

ATTACHMENT 2

SAS Code for Baseline, Hypoglycemia, and Retinopathy Progression Tabulations from EDIC Adolescent Four Year Retinopathy, Nephropathy and Hypoglycemia Datasets in the NIDDK Repository

```
/* ***** /
/*
/* Program: R:\05_Users\Norma\EDIC\Adolescents\getdata.sas
/* Author: Norma Pugh
/* Date: 15 October 07
/* Purpose: Create SAS datasets from the SAS transport files provided to the repository.
/* ***** /
/* NEW LOCATION FOR EDIC SAS FILES */
libname newlib 'R:\05_Users\Norma\EDIC\Adolescents';

/* ORIGINAL LOCATION OF EDIC TRANSPORT FILES */
filename file1 'R:\03_Data_And_Tools\Studies\DCCT-EDIC\NEW-Studies\4-year-
adolescent\edicado4.xpt';
filename file2 'R:\03_Data_And_Tools\Studies\DCCT-EDIC\NEW-Studies\4-year-
adolescent\edad4hyp.xpt';
filename file3 'R:\03_Data_And_Tools\Studies\DCCT-EDIC\NEW-Studies\4-year-
adolescent\edad4adl.xpt';

/* CREATE THE DATASETS */
proc cimport library=newlib infile=file1; run;
proc cimport library=newlib infile=file2; run;
proc cimport library=newlib infile=file3; run;
```

```

/*****/
/*
/* Program: R:\05_Users\Norma\EDIC\Adolescents\verify_trt.sas
/* Author: Norma Pugh
/* Date: 24 October 07
/* Purpose: Verify that treatment assignment matches DCCT Baseline dataset.
/*****/
/*****/
/* Libnames and formats */
/*****/
libname trt 'R:\03_Data_And_Tools\Studies\DCCT-EDIC\DCCT-
EDIC_FINAL\DCCT\Analyses\Baseline\SAS_DATA';
libname data 'R:\05_Users\Norma\EDIC\Adolescents';

%include 'R:\03_Data_And_Tools\Studies\DCCT-EDIC\NEW-Studies\4-year-
adolescent\fmt_ado4.sas';
proc format; value yesno 0='0=no' 1='1=yes'; run;

/*****/
/* Get treatment assignment from DCCT baseline */
/*****/
data trt;
  set trt.baseline;
  keep mask_pat group;
  rename group=baseline_group;
run;

proc sort data=trt; by mask_pat; run;

/*****/
/* Get Adolescent dataset treatment assignments */
/*****/
data temp1;
  set data.edicado4;
  keep mask_pat group;
  rename group=temp1_group;
run;
proc sort data=temp1; by mask_pat; run;
proc freq data=temp1; tables temp1_group / missing; title'edicado4'; run;

data temp2;
  set data.edad4hyp;
  keep mask_pat group;
  rename group=temp2_group;
run;
proc sort data=temp2; by mask_pat; run;
proc freq data=temp2; tables temp2_group / missing; title'edad4hyp'; run;

data temp3;
  set data.edad4ad1;
  keep mask_pat group;
  rename group=temp3_group;
run;
proc sort data=temp3; by mask_pat; run;
proc freq data=temp3; tables temp3_group / missing; title'edad4ad1'; run;

```



```
/* Merge project data with treatment assignment */
data table1; merge temp1(in=x1) trt(in=x2); by mask_pat; if x1 &
baseline_group^=temp1_group; run;
data table2; merge temp2(in=x1) trt(in=x2); by mask_pat; if x1 &
baseline_group^=temp2_group; run;
data table3; merge temp3(in=x1) trt(in=x2); by mask_pat; if x1 &
baseline_group^=temp3_group; run;
```

```

/*****/
/*
/* Program: R:\05_Users\Norma\EDIC\Adolescents\table1.sas
/* Author: Norma Pugh
/* Date: 24 October 07
/* Purpose: Replicate Table 1 results from The Journal of Pediatrics article: Beneficial
/*           effects of intensive therapy of diabetes during adolescence: Outcomes after
/*           the conclusion of the Diabetes Control and Complications Trial (DCCT).
/*           (2001)
/*****/
/*****/
/* Libnames and formats */
/*****/
libname data 'R:\05_Users\Norma\EDIC\Adolescents';
libname trt 'R:\03_Data_And_Tools\Studies\DCCT-EDIC\DCCT-
EDIC_FINAL\DCCT\Analyses\Baseline\SAS_DATA';

%include 'R:\03_Data_And_Tools\Studies\DCCT-EDIC\NEW-Studies\4-year-
adolescent\fmt_ado4.sas';
proc format; value yesno 0='0=no' 1='1=yes'; run;

/*****/
/* Get primary cohort assignment */
/*****/
data trt;
  set trt.baseline;
  keep mask_pat rebase;
run;

proc sort data=trt; by mask_pat; run;

/*****/
/* Get Table 1 population & variables */
/*****/
data table1;
  set data.edicado4(where=(hba1c>.));

  /* Women pregnant during EDIC (%) */
  length preg_dur_edic 4;
  if tot_preg>0 then preg_dur_edic=1; else preg_dur_edic=0;

  /* Labels */
  label preg_dur_edic = 'Pregnant during EDIC (1=y/0=n)';
run;

proc sort data=table1; by mask_pat; run;

/*****/
/* Merge table data with primary cohort assignment */
/*****/
data table1; merge table1(in=x1) trt(in=x2); by mask_pat; if x1 & x2; run;

/*****/
/* Table 1 Replication Analysis */
/*****/
title'Table 1: Characteristics of EDIC patients who were adolescents at entry into the
DCCT';

```

```
title2'and who had an HbA1c at year 4 of EDIC';
title3'Categorical Counts & p-values'; run;

proc freq data=table1;
  tables group;
run;

proc freq data=table1;
  tables group*(retbase sex);
run;

proc freq data=table1(where=(sex='F'));
  tables group*preg_dur_edic;
run;

proc freq data=table1;
  tables group*pregnant;
run;

proc freq data=table1;
  tables group*obinsreg / chisq;
  format obinsreg insfmt.;
run;

proc freq data=table1;
  tables group*sbgm_4;
run;

title2'Quantitative Means & Standard Deviations'; run;

proc sort data=table1; by group; run;
proc means data=table1 n mean std;
  by group;
  var att_age att_dur in_study std_ins hba1c;
run;
```

```

/*****/
/*
/* Program: R:\05_Users\Norma\EDIC\Adolescents\table2.sas
/* Author: Norma Pugh
/* Date: 24 October 07
/* Revised: 02 December 07 to update calculations for event rates per 100 years. DCC
/* responded to questions and provided the appropriate formula on 30NOV2007.
/* Purpose: Replicate Table 2 results from The Journal of Pediatrics article: Beneficial
/* effects of intensive therapy of diabetes during adolescence: Outcomes after
/* the conclusion of the Diabetes Control and Complications Trial (DCCT).
/* (2001)
/*****/
/*****/
/* Libname and formats */
/*****/
libname data 'R:\05_Users\Norma\EDIC\Adolescents';

%include 'R:\03_Data_And_Tools\Studies\DCCT-EDIC\NEW-Studies\4-year-
adolescent\fmt_ado4.sas';
proc format; value yesno 0='0=no' 1='1=yes'; run;

/*****/
/* Get Table 2 population & variables. */
/*****/
proc sort data=data.edicado4(where=(hba1c>.)) out=hba1c; by mask_pat; run;
proc sort data=data.edad4hyp out=dcct; by mask_pat; run;
proc sort data=data.edad4adl(where=(teen=1)) out=edic; by mask_pat; run;

data table2_dcct; set dcct(in=x1);
keep mask_pat group danycs dtcs dcsday danyra dtra draday;
run;

data table2_edic; /* As for Table 1, only keep patients with HbA1c measured at EDIC
year 4 */
merge edic(in=x1) hba1c(in=x2 keep=mask_pat);
by mask_pat;
if x1 & x2;
keep mask_pat group edanycs edcs edanyra edra edmon;
run;

/* Per DCC instructions, re-calculate event rates as follows: */
/* Mean event rate = (total # events / total # followup days)*365.25*100. */
/* Reference: Biostatistical Methods by John Lachin, Chapter 8. */
data table2_dcct;
set table2_dcct;
comarate = (dtcs/dcsday)*365.25*100;
allrate = (dtra/draday)*365.25*100;
label comarate = 'Coma/seizure rate per 100 patient-yrs'
allrate = 'All event req. asst. rate per 100 pt-yrs';
drop dtcs dcsday dtra draday;
run;

data table2_edic;
set table2_edic;
comarate = (edcs/edmon)*365.25*100;
allrate = (edra/edmon)*365.25*100;

```

```

label comarate = 'Coma/seizure rate per 100 patient-yrs'
      allrate = 'All event req. asst. rate per 100 pt-yrs';
drop edcs edra edmon;
run;

/*****/
/* Table 2 Replication Analysis */
/*****/
title 'Table 2: Summary of rates of hypoglycemia in former DCCT adolescents';
title2 'during the DCCT and the first 4 years of EDIC';
title3 'DCCT Results';

proc freq data=table2_dcct;
  tables group;
run;

proc freq data=table2_dcct;
  tables group*danycs / cmh1;
  format danycs yesno.;
run;

proc means data=table2_dcct n mean;
  class group;
  var comarate;
run;

proc freq data=table2_dcct;
  tables group*danyra / cmh1;
  format danyra yesno.;
run;

proc means data=table2_dcct n mean;
  class group;
  var allrate;
run;

title3 'EDIC Results';

proc freq data=table2_edic;
  tables group;
run;

proc freq data=table2_edic;
  tables group*edanycs / cmh1;
  format edanycs yesno.;
run;

proc means data=table2_edic n mean;
  class group;
  var comarate;
run;

proc freq data=table2_edic;
  tables group*edanyra / cmh1;

```

```
format edanyra yesno.;  
run;  
  
proc means data=table2_edic n mean;  
class group;  
var allrate;  
run;
```

```

/*****/
/*
/* Program: R:\05_Users\Norma\EDIC\Adolescents\table3.sas
/* Author: Norma Pugh
/* Date: 24 October 07
/* Purpose: Replicate Table 3 results from The Journal of Pediatrics article: Beneficial
/* effects of intensive therapy of diabetes during adolescence: Outcomes after
/* the conclusion of the Diabetes Control and Complications Trial (DCCT).
/* (2001)
/*****/
/*****/
/* Libname and formats */
/*****/
libname data 'R:\05_Users\Norma\EDIC\Adolescents';

%include 'R:\03_Data_And_Tools\Studies\DCCT-EDIC\NEW-Studies\4-year-
adolescent\fmt_ado4.sas';
proc format; value yesno 0='0=no' 1='1=yes'; run;

/*****/
/* Get Table 3 population & variables */
/*****/
data table3;
set data.edicado4;
run;

/*****/
/* Table 3 Replication Analysis */
/*****/
title'Table 3: Further progression of at least 3 steps in retinopathy level in former
DCCT';
title2'adolescents between DCCT closeout and the EDIC 4-year visit';
title3'Categorical Counts & Percentages'; run;

proc freq data=table3;
tables group*subgrpr / norow nocol nopct;
format subgrpr retstat.;
run;

data table3_tot; set table3; output; subgrpr=99; output; run; /*Add 'total' category*/
proc freq data=table3_tot;
tables subgrpr*group*new3step / nocol nopct;
format subgrpr retstat.;
run;

/* Odds Reduction, 95% CI, p-value */
%MACRO OR(category,out);
proc freq data=table3_tot(where=(subgrpr=&category)) noprint;
tables group*new3step / cmh1;
output out=&out cmh1;
run;

data &out; set &out;
odds_red=(1-(_mhrrc2/_mhrrc1))*100; /*MH Adjusted Col1 RR, MH Adjusted Col2 RR*/
low_ci=(1-(1/l_lgor))*100; /* Lower CL, Logit Adjusted OR */
high_ci=(1-(1/u_lgor))*100; /* Upper CL, Logit Adjusted OR */
pval=round(p_cmhcor,.0001); /* P-value for CMH Nonzero Correlation */

```

```
keep odds_red low_ci high_ci pval;
run;

proc print data=&out; TITLE3"Odds Reduction - &out Stats"; run;
%MEND OR;

%OR(99,All_Patients);
%OR(1,No_Retinop);
%OR(2,Micro_Only);
%OR(3,Mild);
%OR(4,Moderate);

/* Adjusted Odds Reduction, 95% CI, p-value */
proc logistic data=table3 descending;
class group subgrpr;
model new3step = group subgrpr;
title3'Adjusted Odds';
run;
```